NAG Library Routine Document

S17AVF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S17AVF returns an array of values of the Airy function, $\mathrm{Bi}(x)$.

2 Specification

```
SUBROUTINE S17AVF (N, X, F, IVALID, IFAIL)

INTEGER            N, IVALID(N), IFAIL
REAL (KIND=nag_wp) X(N), F(N)
```

3 Description

S17AVF evaluates an approximation to the Airy function $\mathrm{Bi}(x_i)$ for an array of arguments x_i, for $i = 1, 2, \ldots, n$. It is based on a number of Chebyshev expansions.

For $x < -5$,

$$\mathrm{Bi}(x) = \frac{a(t)\cos z + b(t)\sin z}{(-x)^{1/4}},$$

where $z = \frac{\pi}{4} + \frac{2}{3}\sqrt{-x^3}$ and $a(t)$ and $b(t)$ are expansions in the variable $t = -2\left(\frac{5}{x}\right)^3 - 1$.

For $-5 \leq x \leq 0$,

$$\mathrm{Bi}(x) = \sqrt{3}(f(t) + xg(t)),$$

where f and g are expansions in $t = -2\left(\frac{x}{5}\right)^3 - 1$.

For $0 < x < 4.5$,

$$\mathrm{Bi}(x) = e^{11x/8}y(t),$$

where y is an expansion in $t = 4x/9 - 1$.

For $4.5 \leq x < 9$,

$$\mathrm{Bi}(x) = e^{5x/2}v(t),$$

where v is an expansion in $t = 4x/9 - 3$.

For $x \geq 9$,

$$\mathrm{Bi}(x) = \frac{e^z u(t)}{x^{1/4}},$$

where $z = \frac{2}{3}\sqrt{x^3}$ and u is an expansion in $t = 2\left(\frac{18}{z}\right) - 1$.

For $|x| <$ ***machine precision***, the result is set directly to $\mathrm{Bi}(0)$. This both saves time and avoids possible intermediate underflows.

For large negative arguments, it becomes impossible to calculate the phase of the oscillating function with any accuracy so the routine must fail. This occurs if $x < -\left(\frac{3}{2\epsilon}\right)^{2/3}$, where ϵ is the ***machine precision***.

For large positive arguments, there is a danger of causing overflow since Bi grows in an essentially exponential manner, so the routine must fail.

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

5 Parameters

1: N – INTEGER *Input*

On entry: n, the number of points.

Constraint: $\mathrm{N} \geq 0$.

2: X(N) – REAL (KIND=nag_wp) array *Input*

On entry: the argument x_i of the function, for $i = 1, 2, \ldots, \mathrm{N}$.

3: F(N) – REAL (KIND=nag_wp) array *Output*

On exit: $\mathrm{Bi}(x_i)$, the function values.

4: IVALID(N) – INTEGER array *Output*

On exit: IVALID(i) contains the error code for x_i, for $i = 1, 2, \ldots, \mathrm{N}$.

IVALID$(i) = 0$
No error.

IVALID$(i) = 1$
x_i is too large and positive. F(i) contains zero. The threshold value is the same as for IFAIL $= 1$ in S17AHF, as defined in the Users' Note for your implementation.

IVALID$(i) = 2$
x_i is too large and negative. F(i) contains zero. The threshold value is the same as for IFAIL $= 2$ in S17AHF, as defined in the Users' Note for your implementation.

5: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, at least one value of X was invalid.
Check IVALID for more information.

IFAIL = 2

On entry, N = ⟨*value*⟩.
Constraint: N ≥ 0.

7 Accuracy

For negative arguments the function is oscillatory and hence absolute error is the appropriate measure. In the positive region the function is essentially exponential-like and here relative error is appropriate. The absolute error, E, and the relative error, ϵ, are related in principle to the relative error in the argument, δ, by

$$E \simeq \left|x\mathrm{Bi}'(x)\right|\delta, \epsilon \simeq \left|\frac{x\mathrm{Bi}'(x)}{\mathrm{Bi}(x)}\right|\delta.$$

In practice, approximate equality is the best that can be expected. When δ, ϵ or E is of the order of the ***machine precision***, the errors in the result will be somewhat larger.

For small x, errors are strongly damped and hence will be bounded essentially by the ***machine precision***.

For moderate to large negative x, the error behaviour is clearly oscillatory but the amplitude of the error grows like amplitude $\left(\frac{E}{\delta}\right) \sim \frac{|x|^{5/4}}{\sqrt{\pi}}$.

However the phase error will be growing roughly as $\frac{2}{3}\sqrt{|x|^3}$ and hence all accuracy will be lost for large negative arguments. This is due to the impossibility of calculating sin and cos to any accuracy if $\frac{2}{3}\sqrt{|x|^3} > \frac{1}{\delta}$.

For large positive arguments, the relative error amplification is considerable:

$$\frac{\epsilon}{\delta} \sim \sqrt{x^3}.$$

This means a loss of roughly two decimal places accuracy for arguments in the region of 20. However very large arguments are not possible due to the danger of causing overflow and errors are therefore limited in practice.

8 Further Comments

None.

9 Example

This example reads values of X from a file, evaluates the function at each value of x_i and prints the results.

9.1 Program Text

```
    Program s17avfe

!     S17AVF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, s17avf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                           :: i, ifail, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable   :: f(:), x(:)
```

```
      Integer, Allocatable             :: ivalid(:)
!     .. Executable Statements ..
      Write (nout,*) 'S17AVF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Write (nout,*)
      Write (nout,*) '     X           F          IVALID'
      Write (nout,*)

      Read (nin,*) n

      Allocate (x(n),f(n),ivalid(n))

      Read (nin,*) x(1:n)

      ifail = 0
      Call s17avf(n,x,f,ivalid,ifail)

      Do i = 1, n
        Write (nout,99999) x(i), f(i), ivalid(i)
      End Do

99999 Format (1X,1P,2E12.3,I5)
    End Program s17avfe
```

9.2 Program Data

```
S17AVF Example Program Data

7

-10.0 -1.0 0.0 1.0 5.0 10.0 20.0
```

9.3 Program Results

```
 S17AVF Example Program Results

      X           F          IVALID

   -1.000E+01  -3.147E-01    0
   -1.000E+00   1.040E-01    0
    0.000E+00   6.149E-01    0
    1.000E+00   1.207E+00    0
    5.000E+00   6.578E+02    0
    1.000E+01   4.556E+08    0
    2.000E+01   2.104E+25    0
```